

11



PULSADOR




RESISTENCIA DE 10 KILO OHMIOS



RESISTENCIA DE 220 OHMIOS



POTENCIÓMETRO



PANTALLA LCD

BOLA DE CRISTAL

CREAR UNA BOLA DE CRISTAL PARA QUE LE HABLE DE SU FUTURO

Descubra: [Pantallas LCD](#), [instrucciones swith/case](#), [random\(\)](#)


Tiempo: 1HORA

Proyectos en los que se basa: [1,2,3](#)

Nivel: **alto**

La bola de cristal le puede ayudar a "adivinar" el futuro. Le hace una pregunta a la bola que todo lo sabe y a continuación deberá de mover la para obtener la respuesta. Las respuestas estarán guardas con anterioridad, pero podrá redactarlas como más le guste. Se usará Arduino para escoger una respuesta de un total de 8 respuestas guardadas. El sensor de inclinación que se incluye en el kit imita el movimiento de la bola cuando se frota para obtener las respuestas.

La pantalla LCD se puede usar para mostrar caracteres alfanuméricos. El de este kit dispone de 16 columnas y de 2 filas, para un total de 32 caracteres. Su montaje sobre la placa de circuito impreso incluye un gran número de conexiones. Estos terminales se utilizan para la alimentación y comunicación, además de indicar lo que tiene que escribir sobre la pantalla, pero no es necesario conectar todos sus terminales, algunos se dejan al aire. Ver la figura número 3 (esquema eléctrico) para ver los terminales que se conectan al circuito.



Todos los terminales de la pantalla LCD. No se conectan todos.

Figura 1

MONTANDO EL CIRCUITO




Figura 2




Figura 3

En este esquema la disposición de los terminales de la pantalla LCD no coinciden con el orden físico en la figura 2 (montaje sobre la placa de pruebas). En un esquema, los terminales se reorganizan de una manera lógica de forma que el esquema se pueda leer con una mayor claridad. Esto puede causar una pequeña confusión a los principiantes hasta que realicen el montaje del proyecto.

El circuito no es complicado de montar, pero existen un montón de cables que conectar. Prestar atención cuando se realicen todas las conexiones con los cables, asegurarse que todo está bien conectado.

- 1 Conectar los cables de alimentación y masa tal y como se muestra en esta ilustración, en un extremo de la placa de pruebas.
- 2 Colocar el sensor de inclinación en la placa de pruebas y conectar uno de sus terminales a +5V. Conectar el otro terminal a masa y a través de una resistencia de 10K ohmios, el punto de central de unión de esta resistencia y el terminal del sensor se unen al pin número 6 de la placa Arduino. Se configurará como una entrada digital, tal y como se ha realizado en proyectos anteriores.
- 3 El terminal de selección de registro (**RS**) de la pantalla LCD controla cuando los caracteres aparecerán sobre la pantalla. El terminal de lectura/escritura (read/write) (**R/W**) coloca la pantalla en modo de lectura o de escritura. En este proyecto se usará el modo de escritura. El terminal de activación (enable) (**EN**) le dice a la pantalla que va a recibir una instrucción. Los terminales de datos (**D0-D7**) se utilizan para enviar caracteres de datos a la pantalla. Solo se usarán 4 (**D4-D7**) de estos 8 terminales. Finalmente, la pantalla dispone de un terminal a través del cual se puede ajustar el contraste de la misma (**VO**). Se usará un potenciómetro para realizar este control del contraste.
- 4 La librería de la pantalla LCD que se incluye con el software de Arduino maneja toda la información de estos terminales y simplifica el proceso de escribir software para mostrar los caracteres. Los dos terminales externos del LCD (**Vss** y **LED-**) hay que conectarlos a masa. También se conecta a masa el terminal **R/W**, el cual coloca al LCD en modo de escritura. El terminal de alimentación del LCD (**Vcc**) se conecta directamente a +5V. El terminal **LED+** de la pantalla se conecta a través de una resistencia de 220V también a +5V.
- 5 Conectar: El terminal digital 2 de Arduino al terminal **D7** del LCD, el terminal digital 3 al **D6** del LCD, el terminal digital 4 al **D5** del LCD y el terminal digital 5 al **D4** del LCD. Todos son terminales de datos que le dicen a la pantalla que carácter debe de mostrar.
- 6 Conectar el terminal **EN** (activación) de la pantalla al terminal 11 de Arduino. El terminal **RS** de la pantalla se conecta al terminal 12 de Arduino. Este terminal habilita la escritura del LCD.
- 7 Colocar el potenciómetro en la placa de pruebas, conectando uno de sus terminales extremos a la alimentación de 5V y el otro a masa. El terminal central debe de conectarse al terminal **VO** de la pantalla LCD. Este potenciómetro permitirá cambiar el contraste de la pantalla.

EL CÓDIGO

Configurar la librería LiquidCrystal

Primero, es necesario importar la librería **LiquidCrystal**. A continuación, se inicializa esta librería, de la misma forma que se hizo con la librería del servomotor en el proyecto número 5 (indicador del estado de ánimo), indicando en el programa que terminales se van a usar para la comunicación con Arduino.

Una vez que se ha configurado la librería, es el momento de crear algunas variables y constantes. Crear una constante para guardar el número del pin de Arduino al que se conecta el sensor de inclinación (**PinDelSensor**), una variable que almacene el estado actual del de sensor (**EstadoDelSensor**), otra variable para guardar el estado previo del sensor (**EstadoPreviodelSensor**), y una más (**Contestar**) para elegir la respuesta que se mostrará en la pantalla LCD.

Definir el tamaño de la pantalla

Configurar el pin de Arduino al que se conecta el sensor de inclinación como una entrada con la instrucción **pinMode()** dentro de la función **setup()**. Iniciar la librería LCD e indicarle el tamaño de la pantalla (16 columnas por 2 filas).

Mover el cursor y escribir el primer texto

Ahora es el momento de escribir un pequeño texto de introducción a modo de pantalla de bienvenida antes las 8 respuestas posibles. La función **print()** escribe en la pantalla LCD. Se va a escribir el texto "**Preguntame**" en la línea superior de la pantalla. El cursor se coloca automáticamente al comienzo de la línea superior y en la primera columna.

Con el fin de escribir en la siguiente línea, hay que indicarle a la pantalla donde se debe de colocar el cursor (**setCursor(0,1)**). Las coordenadas de la primera columna y de la segunda línea son 0,1 (recuerda que en los ordenadores las coordenadas 0,0 se refiere a la primera columna de la primera fila). Usar la función **lcd.setCursor()** para mover el cursor a la primera columna de la segunda fila y a continuación decirle que escriba la frase "**Bola de Cristal**".

Ahora, cuando comience el programa, aparecerá en la pantalla LCD "**Preguntame Bola de Cristal**".

Dentro de la función **loop()**, primeramente se verifica el estado del sensor de inclinación, y a continuación se coloca el valor leído dentro de la variable **EstadoDelSensor**.

Cooger una respuesta aleatoria

Usar la instrucción **if()** para determinar si el sensor está en una posición diferente a como estaba en un estado previo. Si es diferente a como estaba antes y además en estos momentos está en estado LOW (bajo), entonces es el momento de generar una respuesta aleatoria. La función **random()** devuelve un número dentro del rango que se indica dentro del argumento. Para comenzar, el programa guarda 8 posibles respuestas para la bola de cristal. Siempre que la instrucción **random(8)** es ejecutada, se obtendrá cualquier número comprendido entre 0 y 7. Este número se guarda dentro de la variable **Contestar**.

```
1 #include <LiquidCrystal.h>
2 LiquidCrystal lcd(12, 11, 5, 4, 3, 2);
```

```
3 const int PindelSensor = 6;
4 int EstadodelSensor = 0;
5 int EstadoPreviodelSensor = 0;
6 int Contestar;
```

```
7 void setup() {
8   lcd.begin(16, 2);
9   pinMode(PindelSensor, INPUT);
```

```
10  lcd.print("Preguntame");
```

```
11  lcd.setCursor(0, 1);
12  lcd.print("Bola de Cristal");
13 }
```

```
14 void loop() {
15   EstadodelSensor = digitalRead(PindelSensor);
```

```
16  if(EstadodelSensor != EstadoPreviodelSensor) {
17    if(EstadodelSensor == LOW) {
18      Contestar = random(8);
```

LCD library reference
arduino.cc/libraries

Random reference
arduino.cc/random

Limpiar la pantalla con la función `lcd.clear()`. Estas instrucciones también mueven el cursor a la posición inicial 0,0; la primera columna de la primera fila del LCD. Escribir el texto "La bola dice" en pantalla y mover el cursor a la línea siguiente.

Predecir el futuro

La instrucción `switch()` ejecuta diferentes partes del código dependiendo del valor que tenga esta instrucción. Cada una de estas partes del código se llama mediante una instrucción `case switch()` la cual verifica el valor de la variable `Contestar` (valor como argumento de la instrucción); de esta manera cualquier valor que se genera dentro de la variable `Contestar` determina cual de las instrucciones `case` (de 1 a 7) será ejecutada.

Dentro de las instrucciones `case`, el código será el mismo, pero los mensajes que muestre en pantalla serán diferentes. Por ejemplo, en `case 0` el código hace que a través de la función `lcd.print("Si")` aparezca el texto de "Si" en la pantalla. Después de la función `lcd.print()` hay otro comando: `break`, que le dice a Arduino donde se localiza el final de la última instrucción `case`. Cuando se ejecuta `break`, el programa salta al final de la última sentencia de la instrucción `switch`, la cual siempre será `}`. Para comenzar se van a crear un total de 8 instrucciones `case`. Cuatro de las respuestas serán positivas, dos serán negativas, una respuesta que no tiene ni idea y otra respuesta que vuelva a preguntar de nuevo.

Lo último que hay que hacer al final del programa, dentro de la función `loop()`, es asignar el valor de la variable `EstadoDelSensor` a la variable `EstadoPreviodelSensor`. Esto posibilita que se pueda realizar un seguimiento del cambio de estado (de 0 a 1 ó de 1 a 0) en el sensor de inclinación la próxima vez que el programa se ejecute.

```

19    lcd.clear();
20    lcd.setCursor(0, 0);
21    lcd.print("La bola dice:");
22    lcd.setCursor(0, 1);

23    switch(Contestar){
24        case 0:
25            lcd.print("Si");
26            break;
27        case 1:
28            lcd.print("Es probable");
29            break;
30        case 2:
31            lcd.print("Ciertamente");
32            break;
33        case 3:
34            lcd.print("Buenas perspectivas");
35            break;
36        case 4:
37            lcd.print("No es seguro");
38            break;
39        case 5:
40            lcd.print("Pregunta de nuevo");
41            break;
42        case 6:
43            lcd.print("Ni idea");
44            break;
45        case 7:
46            lcd.print("No");
47            break;
48    }
49 }
50 }

```

Switch Case reference
arduino.cc/switchcase

```

51 EstadoPreviodelSensor = EstadodelSensor;
52 }

```


COMO SE UTILIZA

Para usar la bola de cristal primero hay que darle alimentación a la placa Arduino. Comprobar si en la pantalla del LCD aparece el texto "**Preguntame**" y en la segunda línea "**Bola de Cristal**". Si no puede ver bien los textos deberá ajustar el potenciómetro para variar el contraste hasta que los caracteres se vean correctamente.

Realizar una pregunta a la bola de cristal y a continuación mover hacia arriba y hacia abajo el circuito para que el sensor de inclinación pueda detectar estas variaciones. De esta manera obtendrá una respuesta a su pregunta. Si la respuesta no le satisface, puede preguntar de nuevo.



Intente añadir sus propias respuestas a las instrucciones **print()**, pero teniendo en mente que solo es posible usar 16 caracteres por línea. Asegurarse cuando se añaden nuevas instrucciones case switch, que se ha ajustado el número de opciones que la variable Contestar va a almacenar; recordar que esto se define mediante la instrucción **random(8)**, en este caso 8 respuestas. Si se añaden dos nuevas respuestas habrá que aumentar el número 8 a 10 y añadir dos nuevas instrucciones case switch.



La pantalla LCD trabaja cambiando las propiedades eléctricas de un líquido que se encuentra montado entre cristales polarizados. El cristal solo permite que un tipo de luz pase a través de él. Cuando el líquido entre los cristales está cargado se vuelve semi sólido de manera que puede representar caracteres al bloquear la luz que incide sobre él.



Las funciones descritas aquí para cambiar el texto que aparece en la pantalla LCD son muy simples. Una vez que tenga claro como funciona puede estudiar la librería de este LCD para descubrir nuevas funciones. Puede intentar que el texto se mueva a lo largo de la pantalla o que se actualice continuamente. Para saber más sobre como trabajar con la librería de la pantalla LCD visitar: arduino.cc/lcd

Una pantalla LCD permite mostrar textos usando la librería LiquidCrystal. Con las instrucciones switch...case controla el flujo de partes del programa al poder comparar una variable con determinados valores.

